
1. Free Pascal

Zasady pisania programów w języku Pascal dla kompilatora Free Pascal nie różnią
się zasadniczo od języka Pascal, znanego np. z popularnego Turbo Pascala (DOS) czy Delphi
(Windows).  Poniżej  opisałem  skrótowo  podstawowe  zasady  programowania  w  Pascalu
i najczęściej używane konstrukcje, z myślą o początkującym Czytelniku. Osoby, które miały
już styczność z językiem Pascal mogą śmiało pominąć ten rozdział. 

Dlaczego język Pascal?

Język  Pascal  jest  obecnie  przez  programistów  nieco  zapomniany,  ale  nadal  jest
użyteczny.  Chociaż uznawany jest za język wysokiego poziomu (tj.  pozwala na używanie
konstrukcji  dość  abstrakcyjnych  pisanych  w  języku  zbliżonym  do  naturalnego,  w  tym
przypadku  angielskiego),  jest  bardzo  elastyczny,  pozwalając  na  programowanie
niskopoziomowych procedur  (tj.  praktycznie  w kodzie  maszynowym za  pomocą wstawek
asemblerowych). Poza tym jest względnie łatwy do nauki. Dlatego moim zdaniem znakomicie
nadaje  się  tak  do  pisania  programów  desktopowych  (na  PC)  jak  i  do  programowania
systemów wbudowanych (embedded). 

Obecnie  wśród programistów systemów wbudowanych najbardziej  popularny jest
język C, także w odmianie stosowanej w popularnym Arduino. Język Pascal jest nieco mniej
elastyczny i bardziej rygorystyczny, szczególnie w operowaniu na zmiennych. To jest moim
zdaniem jego zaleta.  Ma też swoje ograniczenia – o czym w dalszej części  tekstu.  Wiele
zależy od przyzwyczajeń Czytelnika, ale myślę, że warto spróbować!

Bloki instrukcji, stałe, zmienne, podstawowe typy

W języku Pascal instrukcje powinny być zamknięte  między słowami kluczowymi
begin i end. W tego typu bloku można używać zarówno stałych (ich wartości raz określonych



w programie nie można już zmieniać) jak i zmiennych. Pascal wymusza jednak deklaracje
stałych i zmiennych przed blokiem instrukcji,  w których będą wykorzystywane. Należy je
oznaczyć  słowami  kluczowymi  const (dla  stałych)  i  var (dla  zmiennych).  Przypisanie
wartości do stałych i zmiennych przypomina równania matematyczne z tą różnicą, że wartości
stałych przypisywane są za pomocą znaku =, natomiast zmiennych przy pomocy :=. Ponadto
w  języku  Pascal  można  operować  na  różnych  typach  stałych  i  zmiennych,  z  których
podstawowe to Byte (bajt, liczba od 0 do 255), Integer (liczba całkowita) oraz String (łańcuch
znaków). Istnieją również typy liczbowe zmiennoprzecinkowe (np. Float, Double, które tutaj
pominę)  czy  bardziej  specyficzne  (por.  kolejne  podrozdziały).  Separatorem instrukcji  jest
średnik.  W  kodzie  można  dodawać  komentarze,  rozpoczynające  się  od  znaków  // i
obejmujące tekst do końca linii, albo zamknięte między {} niezależnie od położenia w kodzie
źródłowym.

Przykład:

const
  a = 100; //deklaracja stałej
var //poniżej deklaracje zmiennych
  b: Integer; 
  c: String;
begin //początek bloku instrukcji
  b:=a; //instrukcja przypisania do zmiennej b wartości stałej a
  b:=b-1; //po wykonaniu tej instrukcji zmienna b będzie równa 99
  c:=’Free Pascal dla AVR’;
end. //koniec bloku instrukcji (także koniec programu)

Typ logiczny Boolean, instrukcje warunkowe, pętle

W języku Pascal  istnieje  typ  logiczny zwany  Boolean przyjmujący  wartość  false
(fałsz) lub true (prawda). Pozwala on np. na konstruowanie rozmaitych warunków logicznych
w  kodzie.  Ponadto  Pascal  umożliwia  pisanie  instrukcji  warunkowych  między  słowami
kluczowymi  if [warunek]  then [instrukcje  jeśli  warunek  spełniony]  else [instrukcje  jeśli
warunek niespełniony]. 

Przykład:

var //poniżej deklaracje zmiennych
  b: Boolean; 
  a, c: Integer; 

[...] //jakieś instrukcje

b:=a>0; //zmienna b przyjmie wartość true jeśli a>0
if (b=true) then
begin
  c:=c+1; //instrukcja wykonana jeśli b=true (tj. a>0)
end
else
begin
  c:=c-1; //instrukcja wykonana jeśli b=false (tj. a<=0)
end;

W  przypadku  bardziej  rozbudowanych  warunków  stosuje  się  konstrukcję  case
[zmienna] of [warunki: instrukcje] end;

Przykład:

case a of
  1: b:=b+1; //instrukcja wykonana jeśli a=1
  2: b:=b-1; //instrukcja wykonana jeśli a=2
  else b:=0; //instrukcja wykonana w pozostałych przypadkach
end;

Można stosować kilka rodzajów pętli: z licznikiem, warunkiem oraz nieskończone.
W przypadku prostej pętli z licznikiem dana zmienna inkrementowana (tj. zwiększana) jest
określoną ilość razy.

Przykład:

for i:=1 to 10 do //inkrementacja zmiennej i w zakresie od 1 do 10
begin
  //instrukcje
end;

Z  kolei  pętla  z  warunkiem  wykonywana  jest  tak  długo  jak  dany  warunek  jest
spełniony. Szczególnym przypadkiem jest tutaj pętla nieskończona, dla której warunek jest
zawsze spełniony.

Przykład:

while true do
begin
  //instrukcje wykonywane w nieskończoność
end;

Tablice, typy rekordowe 

Język  Pascal  pozwala  także  na  operowanie  bardziej  złożonymi  typami  danych.
Należą  do  nich  tablice  oraz  typy  rekordowe.  Tablice  to  nic  innego  jak  uporządkowane
w szeregu dane tego samego typu, przy czym mogą mieć one więcej niż jeden wymiar (dwu-
i wielowymiarowe). Tablice deklaruje się za pomocą słowa kluczowego  array a następnie
określa  się  rozmiar  tablicy  zamknięty  między  znakami  [] po  czym  określa  się  typ  jej
elementów składowych.

Przykład:

var
   x: array[0..99] of Integer; //tablica zawierająca 100 elementów typu 
Integer, dostęp do elementu za pomocą indeksu x[index]

Z  kolei  typy  rekordowe  to  typ  danych  łączących  wiele  (minimum  jedno)  pól
rozmaitego typu. Definiuje się je za pomocą słowa kluczowego type, po czym należy określić
nazwy i typy pól składowych.

Przykład:

type //definicja typu rekordowego
  TMyRecord=record
    a: Integer; //pole typu Integer

    b: String; //pole typu String
  end;

var
  x: TMyRecord; //deklaracja zmiennej x typu TMyRecord
begin
  x.a:=10; //przypisanie wartości pól
  x.b:='Free Pascal dla AVR';
end.

Procedury, funkcje

Procedury i funkcje można traktować jako bloki kodu, którym nadano nazwę, oraz
do których można przekazywać parametry (stałe lub zmienne różnego typu). W języku Pascal
procedura  różni  się  od  funkcji  tym,  iż  funkcja  może  zwracać  wynik  danego  typu,  zaś
procedura jest jedynie blokiem kodu. Ponadto zadeklarowane uprzednio procedury i funkcje
można wywoływać w dalszej części kodu. Deklaracja  procedury rozpoczyna się do słowa
kluczowego  procedure,  po czym w nawiasach należy podać listę parametrów jeśli  są one
potrzebne.  Z  kolei  funkcja  rozpoczyna  się  od  słowa  function,  a  po  ewentualnej  liście
parametrów należy określić typ zwracanej wartości.

Przykład:

function Sum(a, b: integer): integer;
begin
  Sum:=a+b; //funkcja zwraca sumę parametrów a i b
end;

procedure DoSum;
var
  a, b, c: integer;
begin
  a:=10;
  b:=1;
  c:=Sum(a, b);
end;

Typy klasowe

Typy  klasowe  to  bardziej  złożone  zagadnienie  związane  z  programowaniem
obiektowym.  Silnie  upraszczając  łączą  w  sobie  funkcjonalność  typów  rekordowych
i procedur/funkcji. Bywają różnie definiowane, inaczej w Turbo Pascalu, nieco inaczej w Free
Pascalu/Delphi.  Podobnie jak typy rekordowe mogą zawierać pola,  ale co bardziej  istotne
mogą  także  zawierać  odpowiedniki  procedur/funkcji  zwane  metodami.  Ponadto  zawierają
metody szczególne, zwane konstruktorem i destruktorem wykorzystywane podczas tworzenia
danego obiektu w pamięci oraz jego zwalniania z pamięci komputera. 

Przykład (silnie uproszczony):

program MyProgram;

type
  TMyClass = class //definicja klasy TMyClass
    a: integer; //pole typu Integer
    constructor Create; //rodzaj procedury zwany konstruktorem
  end;

constructor TMyClass.Create;
begin
  a:=1; //przypisanie do pola a wartości podczas tworzenia obiektu typu 
TMyClass
end;

var
  MyClass: TMyClass; //zmienna typu TMyClass
  b: integer;
begin
  MyClass:=TMyClass.Create; //dynamiczne utworzenie obiektu TMyClass
  b:=MyClass.a; //przypisanie do zmiennej b wartości pola a (1)
  MyClass.Free; //zwolnienie pamięci zajmowanej przez zmienną MyClass
end.

Powyższy  przykład  jest  jednocześnie  przykładem  prostego  programu  napisanego
w Pascalu. Wskazuje na to słowo kluczowe program (za nim nazwa programu) w pierwszej
linii  kodu, po czym następuje deklaracja typu klasowego oraz definicja jego konstruktora.
Główny  blok  kodu  zawiera  się  między  słowami  kluczowymi  begin i  end i  musi  być
zakończony kropką.

Moduły, biblioteki

Prędzej  czy  później  zachodzi  potrzeba  wydzielenia  procedur/funkcji
wykorzystywanych  w  różnych  programach  i  zapisania  ich  w  jednym  miejscu.  Pliki,
zawierające  tego  typu  procedury/funkcje  zwane  są  modułami.  Modułów  („unitów”)  nie
można „uruchomić”, ale można z nich korzystać w programach. Moduły muszą się zaczynać
od słowa kluczowego unit po którym należy określić nazwę modułu. Ponadto moduły muszą
mieć  odpowiednią  strukturę  składającą  się  z  sekcji  interface i  implementation. Sekcja
interface powinna  zawierać  deklaracje  funkcji/procedur  (bez  bloku  instrukcji),  natomiast
w sekcji  implementation procedury/funkcje powinny zawierać blok instrukcji (kodu) begin -
end. Kod modułu musi być zakończony słowem kluczowym end oraz kropką.

Przykład:

unit TestUnit;

interface

function Sum(a, b: integer): integer; //deklaracja funkcji

implementation

function Sum(a, b: integer): integer; //kod funkcji
begin
  Sum:=a+b;
end;

end.

Aby funkcja  Sum z powyższego przykładu mogła zostać wywołana w programie,
w  jego  kodzie  należy  dodać  sekcję  uses,  w  której  należy  wymienić  moduły,  z  których
program będzie korzystał.  

Przykład:

program MyProgram;

uses
  TestUnit;

var
  b: integer;
begin
  b:=Sum(10,2); //po wywołaniu funkcji Sum wartość zmiennej b=12
end.

Często moduły grupowane są w większe zbiory zwane bibliotekami. Służą one do
współdzielenia kodu przez różne programy bez potrzeby jego przepisywania oraz do dzielenia
dużych programów na mniejsze logiczne części.

2. Free Pascal dla AVR

Niniejszy rozdział oraz kolejne to dość swobodne tłumaczenie tutoriala  AVR Programming
przygotowanego przez środowisko twórców Free Pascala. Jest on dostępny na stronie https://
wiki.freepascal.org/AVR_Programming.

Moje komentarze i uwagi umieściłem w ramkach podobnych do tej.

Instrukcje  kompilatora  Free  Pascal  (FPC)  służące  do  programowania
mikrokontrolerów AVR, takich jak ATmega czy ATtiny to instrukcje standardowe dla języka
Pascal. Ważną różnicą jest fakt, iż mikrokontrolery te nie posiadają systemu operacyjnego
(więc nie można stosować poleceń systemowych) oraz potrafią odwoływać się bezpośrednio
do sprzętu. Ponadto istnieje  kilka zagadnień, o których warto wiedzieć programując AVR-y.

Przede  wszystkim  można  stosować  jedynie  dwa  typy  plików  źródłowych,  tj.
program i  unit. Ponieważ kod nie jest uruchamiany na żadnym systemie, inne typy plików
źródłowych (library,  package) nie mają zastosowania w programowaniu mikrokontrolerów
AVR.  W  praktyce  kompilator  FPC  konwertuje  kod  pascalowy  do  kodu  asemblera,
a następnie wywołuje narzędzia GNU toolchains dla mikrokontrolerów AVR, tj. korzystając
z asemblera  avr-as i linkera  avr-ld. Efektem są trzy pliki wynikowe: *.bin (binarny), *.elf
(format linkera) i *.hex (format tekstowy).

3. Zmienne

Deklaracji  zmiennych  dokonuje  się  tak  samo jak  w każdych  innych  programach
napisanych  w  Pascalu.  Ponieważ  system  operacyjny  umożliwiający  dynamiczną  alokację
pamięci nie jest dostępny, nie mogą być stosowane typy dynamiczne takie jak AnsiString.

Zmienne liczbowe całkowite

AVR  to  architektura  8-bitowa,  której  natywny  typ  danych  to  Byte (=  uInt8)
o zakresie od 0 do 255 oraz Shortint (= Int8) w zakresie od -128 do 127. Obliczenia i operacje
na tych dwóch typach danych są zawsze najszybsze. Różnica w porównaniu z WinAPI jest
taka, że typ Integer ma 16 bitów o zakresie od -32768 do 32767 (w WinAPI typ Integer jest

32-bitowy). Korzystanie z typów innych niż natywne zwiększy czas obliczeń, jednak może
mieć uzasadnienie, w zależności od zastosowania. 

Dla pętli z iteracją mniejszą niż 256 cykli odpowiednim typem zmiennych jest Byte.
O ile  można  swobodnie  używać zmiennych  typu  Byte i  Integer,  dla  przejrzystości  warto
stosować ich bardziej opisowe równoważniki, np. Int8 (typ 8-bitowy ze znakiem), uInt8 (typ
8-bitowy  bez znaku) czy  Int16 (typ 16-bitowy bez znaku) lub uInt16 (typ 16-bitowy bez
znaku).  Można  także  stosować  typy  32-bitowe  (Int32 i  uInt32)  oraz  64-bitowe  (Int64
i uInt64).

Zmienne liczbowe zmiennoprzecinkowe

Architektura  AVR  nie  posiada  jednostki  zmiennoprzecinkowej  (koprocesora
matematycznego). Typy Float (i Double) nie są obsługiwane ani sprzętowo ani programowo.
Chociaż potencjalnie istnieje możliwość programowej obsługi typów zmiennoprzecinkowych,
wydaje się ona niepraktyczna z uwagi na duże wymagania programowe (obszerna ilość kodu
w pamięci flash, powolność działania). 

Znacznie  mniej  zasobów  wymaga  obsługa  liczb  stałoprzecinkowych,  które
w  większości  przypadków  są  zupełnie  wystarczające  do  rozwiązań  wykorzystujących
mikrokontrolery AVR. Typy stałoprzecinkowe nie są jednak częścią Free Pascala dla AVR
i wymagają zewnętrznego oprogramowania.  Zaimplementowałem je w formie 16-bitowej
(TFix16) w bibliotece UnoLib (https://sourceforge.net/projects/unolib/). Chociaż biblioteka
ta   przeznaczona  jest  dla  Arduino  Uno  i  Atmega328p  to  w  przypadku  innych
mikrokontrolerów  moduł  stałoprzecinkowy  również  powinien  działać  (o  ile  wystarczy
miejsca  w  pamięci  flash).  Jeśli  mimo  wszystko  zaistnieje  konieczność  użycia  liczb
zmiennoprzecinkowych, wówczas można skorzystać z zawartego również w UnoLib modułu
obsługującego liczby zmiennoprzecinkowe pojedynczej precyzji (TFloat32). 

Zmienne łańcuchowe

Jako, że typy AnsiString i  WideString są przydzielane dynamicznie, można używać
tylko łańcuchów znaków o stałej długości.

Przykład:

// passing a string as a procedure variable
procedure UARTSend(const AText: ShortString);
// using a variable with fixed length string
var 
  myString: String[5];

Uwaga:  ciąg  znaków  zadeklarowany  jako  ShortString ma  zawsze  długość  255
znaków. Jeśli ta długość jest zbyt duża, lepiej użyć ciągu o stałej długości, aby nie marnować
ograniczonej  pamięci.  Można używać typu  ShortString tylko wtedy, gdy nie wiadomo jak
długi ostatecznie będzie łańcuch znaków.

Typ PChar działa.

4. Rejestry sprzętowe

Dostęp do rejestrów

Wszystkie  rejestry  konfiguracji  sprzętowej  są  zdefiniowane  w  pliku
konfiguracyjnym danego mikrokontrolera, znajdującym się w źródłach Pascala w katalogu \
rtl\embedded\avr.  Nazwa pliku konfiguracyjnego jest  jednocześnie nazwą mikrokontrolera,
np.  Atmega32.pp dla   Atmega32.  Plik  oprócz  listy  rejestrów  konfiguracyjnych  zawiera
również nazwy  predefiniowanych przerwań sprzętowych. Plik dołączany jest automatycznie
do wszystkich modułów, a więc zmienne i stałe rejestrów są dostępne w każdym miejscu
programu. Mikrokontroler, a co za tym idzie jego plik konfiguracyjny jest definiowany przez
parametr wiersza poleceń –wpdevicename (np. –wpatmega32). 

W  AVRPascalu  pliki  konfiguracyjne  znajdują  się  w  katalogu  \rtl\avr.  Nie  ma
potrzeby używania wiersza poleceń, gdyż program to robi w tle.

Nazwy  rejestrów  sprzętowych  odpowiadają  nazwom  rejestrów  zdefiniowanych
w  nocie  katalogowej  danego  mikrokontrolera.  Dostęp  do  rejestrów  sprzętowych  jest
analogiczny jak w przypadku do zmiennych. Np.:

UDR:= 25;

zapisuje wartość 25 w rejestrze danych UART (UART Data Register). Do operacji na bitach
można używać przesunięć bitowych (shl = „shift left”), np.:

UCSRB:= (1 shl TXEN) or (1 shl RXEN) or (1 shl RXCIE);
UCSRB:= (1 shl 3) or (1 shl 4) or (1 shl 7);
UCSRB:= %10011000; //Bit 3,4 and 7 are set

Powyższe trzy linie w wygenerowanym kodzie są identyczne: bity TXEN, RXEN
i RXCIE są ustawione w USART Control and Status Register B, co powoduje, że UART jest
dostępny do wysyłania i odbierania danych oraz do obsługi przerwań przy odbiorze danych.

Aliasy nazw rejestrów

W celu  abstrakcji  sprzętu,  w  aplikacjach  często  definiuje  się  specyficzne  nazwy
rejestrów konfiguracji  sprzętu, szczególnie dla rejestrów portów. Można to zrobić poprzez
definicję  zmiennej  przy użyciu  modyfikatora  (słowa kluczowego)  absolute.  W poniższym
przykładzie  LEDPort jest  inną nazwą rejestru sprzętowego PortA. Tego typu definicje  nie
zwiększają zużycia pamięci ani nie zmniejszają wydajności.

Przykład:

var
  LEDPort: byte absolute PortA;

const
  LEDPin = 3;

implementation

LEDPort:= LEDPort or (1 shl LEDPin); //LED=on

5. Działania na bitach

Arytmetyka bitowa jest standardem w Free Pascalu, ale ponieważ jest szczególnie
ważna dla  aplikacji  typu  embedded i  dość rzadko używany w aplikacjach  desktopowych,
warto o niej wspomnieć. Dostępne są następujące operatory służące do manipulacji bitami:
Not, And, Or, Xor, Shl i Shr. Przykład:

ByteVal1:= (ByteVal1 and %11110000) or %00001000;

Prefiks  „%”  oznacza  liczbę  binarną  a  „$”  liczbę  szesnastkową.  W  przypadku
wbudowanego asemblera odpowiadają im „0b” i „0x”.

Często  zachodzi  potrzeba  odczytania  lub  zmiany  konkretnego  bitu,  np.  danego
portu  mikrokontrolera.  Można  tego  dokonać  korzystając  z  operacji  bitowych  (jw.)  lub
zdefiniować własny typ danych w taki sposób, aby do poszczególnych bitów odwoływać się
jak do elementów tablicy lub pól rekordu. Przykład:

type
  TBinary = 0..1; // można też użyć typu boolean (true/false)

  TBitArray = bitpacked array [0..7] of TBinary; //deklaracja bajtu jako 
tablicy bitów

  TBitRec=bitpacked record //deklaracja bajtu jako rekordu z polami 
reprezentującymi bity
    Bit0: TBinary;
    Bit1: TBinary;
    Bit2: TBinary;
    Bit3: TBinary;
    Bit4: TBinary;
    Bit5: TBinary;
    Bit6: TBinary;
    Bit7: TBinary;
  end;

var
  PORTBArray: TBitArray absolute PORTB;
  PORTBRec: TBitRec absolute PORTB;
begin
  PORTBArray[0]:=1; //PORTBArray to alias, wskazujący na PORTB, dostęp do 
bitów jak do elementów tablicy
  PORTBRec.Bit7:=1; //PORTBRec to alias, wskazujący na PORTB, dostęp do 
bitów jak do pól rekordu
  //Jeśli PORTB pierwotnie był "pusty", to po tych operacjach odczytywany 
jako bajt będzie miał wartość 129
end.

6. Klasy/obiekty

Typowe klasy są alokowane dynamicznie i dlatego nie są dostępne. Natomiast klasy
statyczne (słowo kluczowe object) mogą być używane do strukturyzowania i wielokrotnego
wykorzystania kodu. Przykład:

type 
  TUART = object
  procedure Init;
  procedure Send(ByteVal:Byte);
end;

var
  UART: TUART;

Użycie:

UART.Init;
UART.Send(125);

Zmienna  obiektowa UART w przypadku  statycznych  typów klasowych jest  już
zainicjowana i gotowa do użycia. 

7. Przerwania

Aby skorzystać  z  przerwań  sprzętowych  należy  skonfigurować  i  włączyć  źródło
przerwania, włączyć obsługę przerwań oraz zdefiniować procedurę obsługi przerwania (ISR),
która  jest  wywoływana  przez  procesor  w  momencie  wystąpienia  danego  przerwania.
Konfiguracja  odbywa się  poprzez  rejestry  sprzętowe.  Do  globalnego  włączenia  przerwań
służy polecenie asemblera SEI, które można wywołać korzystając z wbudowanego asemblera:

asm SEI end;

Globalne wyłączenie obsługi przerwań odbywa się za pomocą polecenia CLI:

asm CLI end;

Uwaga:  Od  wersji  FPC  3.2.0  można  także  włączać  i  wyłączać  przerwania  bez
odwoływania się bezpośrednio do asemblera:

uses
  intrinsics;

procedure unInterruptableCode;
begin
  avr_cli;
  // code that shouldn't be interrupted
  avr_sei;
end;

Powyższy kod może mieć pewien efekt uboczny polegający na włączeniu przerwań,
nawet  jeśli  są  one  pierwotnie  wyłączone.  Aby  temu  zapobiec  należy  wywołać  avr_save
i  zapisać  rejestr  stanu (wyłącza  to  również przerwania),  wykonać kod,  który nie  zostanie
przerwany, a następnie przywróć rejestr stanu:

uses
  intrinsics;

procedure unInterruptableCodeImproved;
var
  prevState: byte;
begin
  prevState := avr_save;
  // code that shouldn't be interrupted
  avr_restore(prevState);
end;

Do  zdefiniowania  procedury  obsługi  przerwania  ISR  należy  używać  nazw
predefiniowanych dla danego mikrokontrolera. Właściwe nazwy przerwań można znaleźć w
modułach  konfiguracyjnych  danego  mikrokontrolera  (np.  Atmega32.pp)  i  należy  je
zastosować  używając  modyfikatora  alias.  Działający  ISR wygląda  następująco  (przykład:
Przerwanie zewnętrzne 0):

procedure ExternalInterrupt0_ISR; alias: 'INT0_ISR'; interrupt; public;
begin
  //Do some stuff...
end;

Sama  nazwa  procedury  może  być  losowo dobrana,  ważna  jest  nazwa,  do  której
odwołuje się alias. Modyfikator alias zmusza kompilator do użycia nazwy INT0_ISR dla tej
procedury. Jeśli nazwa kryjąca się za aliasem pasuje do predefiniowanej nazwy przerwania,
wówczas  procedura  zostanie  automatycznie  zarejestrowana  w tablicy  wektorów przerwań.
Aby  taki  automatyzm  był  możliwy,  procedura  musi  być  widoczna  globalnie,  co  jest
realizowane  za  pomocą  modyfikatora  public.  Jest  to  jednak  potrzebne  tylko  wtedy,  gdy
procedura jest zdefiniowana w pliku źródłowym typu unit. 

Uwaga:  nie  ma  potrzeby  umieszczania  definicji  procedury  obsługi  przerwania
w  sekcji  INTERFACE  modułu,  wystarczy  umieścić  kod  procedury  w  sekcji
IMPLEMENTATION. Jeśli procedura jest umieszczana w pliku typu program, a nie w unit,
wówczas  modyfikator   public  jest  pomijany.  Modyfikator  interrupt zapewnia  zapisanie
i odzyskanie zawartości wszystkich rejestrów oraz rejestru stanu przez ISR, a także umożliwia
dalsze obsługiwanie przerwań. Pominięcie modyfikatora interrupt spowoduje nieprawidłowe
działanie i awarię programu.

8. Wbudowany asembler

W  niektórych  przypadkach  wydajność  programu  można  znacznie  zwiększyć  za
pomocą kilku instrukcji asemblera. Wbudowany asembler umożliwia bardzo wygodny sposób
korzystania z instrukcji niskopoziomowych.

Blok asm

Blok asemblera rozpoczyna się słowem kluczowe asm i kończy słowem kluczowym
end.

asm
  // do some assembly
end;

W  bloku  asemblera  rejestry,  które  są  modyfikowane  w  kodzie  muszą  zostać
wskazane kompilatorowi za słowem kluczowym  end. Dzięki temu  kompilator będzie mógł
poprawnie obsługiwać rejestry. Wynikowy kod jest bardziej efektywny niż ręczny  pushing
rejestrów w sekcji asm. Zmodyfikowane, ale „nieopublikowane” rejestry mogą prowadzić do
nieprawidłowego działania programu.

asm
  ldi r20,35 // r20 modified
  ldi r21,12 // r21 modified
end['r20','r21']; // publish modified registers to compiler

11



Kompilator  zakłada,  że  rejestr  r1  zawsze  ma  wartość  zero.  Jeśli  został
zmodyfikowany np. przez operację mul należy go następnie wyczyścić.

mul r16,r17
movw r1:r0,r17:r16 // result stored in r1 and r0
clr r1 // clear r1

We  wbudowanym  asemblerze  można  uzyskać  dostęp  do  stałych  i  zmiennych
wykorzystywanych w kodzie. Istnieje różnica między zmiennymi lokalnymi (przydzielonymi
do stosu), które są ładowane i przechowywane za pomocą instrukcji LDD i STD (ograniczone
do  64  bajtów)  i  zmiennymi  globalnymi  (alokowane  statycznie),  które  są  ładowane
i przechowywane za pomocą instrukcji LDS i STS. Zmienne przekazywane przez procedury
są  alokowane  na  stosie  i  tym  samym  są  obsługiwane  jak  zmienne  lokalne.  W  takich
przypadkach warto sprawdzić listing kodu w asemblerze  wygenerowany przez  kompilator
(plik z rozszerzeniem .s).

Podczas korzystania ze skoków stosuje się etykiety. Należy je zadeklarować w sekcji
label procedury. Aby to umożliwić należy zastosować dyrektywę kompilatora {$goto on}.

W językach wysokiego poziomu stosowanie etykiet jest niewskazane i zwykle jest
domyślnie  wyłączone.  Jednak  w  przypadku  bloków  asemblerowych  etykiety  są  często
konieczne.

Uwaga:  Rejestry  konfiguracji  peryferiów  zdefiniowane  w  pliku  konfiguracyjnym
danego mikrokontrolera  (np.  „atmega32.pp”)  to zmienne posiadające  adresy w przestrzeni
adresowej pamięci a NIE w przestrzeni I/O. Adresy w przestrzeni I/O są przesunięte o 0x20,
jako, że  rejestry ogólnego przeznaczenia znajdują się na dole przestrzeni adresowej. Zatem
polecenia  I/O  asemblera  takie  jak  in,  out,  sbis,  sbi (i  tak  dalej)  nie  działają  z  nazwami
rejestrów. Warto spojrzeć na mapą pamięci w nocie katalogowej danego mikrokontrolera.

Zamiast in/out można użyć poleceń lds i sts:

lds r16, PORTB //Reads PORTB into register r16

W kodzie asemblerowym mogą być używane definicje bitowe rejestrów konfiguracji
peryferiów.

Jeśli  konieczne  jest  użycie  instrukcji  I/O asemblera,  adresy  rejestrów I/O można
przesunąć „ręcznie” o 32 (=0x20):

//Wait until UART data register is empty
procedure UARTBusyWait; assembler;
asm
  .waitBusy:
  sbis UCSRA+(-32),UDRE
  rjmp .waitBusy
end;

Poniżej przykład procedury opóźniającej z użyciem wbudowanego asemblera, która
powoduje pauzę programu i zajmuje procesor na pewien czas:

unit Wait; //wait.pas

{$mode objfpc}{$H+}
{$goto on}

//Important for using the label directive, the label is needed in the 
assembler section

interface

const

f_CPU = 4000000; //Hertz

procedure Wait10m(Time: Byte);

implementation

// Waitingtime = Time * 10 Milliseconds
procedure Wait10m(Time: Byte);
const
  Faktor = 15 * f_CPU div 1000000;
label
  // Labels, here for the loop, has to be declared explicitly
  outer, inner1, inner2;
var
  tmpByte: byte;
  // In Inline assembler local variables are accesed by the instructions 
LDD 
  // and STD,global variables are accessed by LDS and STS.
begin
  asm
    // asm states inline assembly block until the next END statement
    ldd r20,Time
    // Variables can be accessed, here a local variable
    outer:
      ldi r21, Faktor
    inner1:
    // 640*Faktor= 640*15*1 = 9600 cycles/1MHz
      ldi r22,128
    inner2:
    // 5*128 = 640 cycles
      nop // 1 cycle
      nop // 1 cycle
      dec r22 // 1 cycle
      brne inner2 // 2 cycles in case of branch 
    //one loop in sum 5 cycles
      dec r21
      brne inner1
      dec r20
      brne outer
    end['r20','r21','r22']; //Used registers to be published to compiler
  end; //procedure
end.
//unit

Procedura asm

Istnieje drugi sposób użycia wbudowanego asemblera poprzez zdefiniowanie całej
procedury jako „asemblerowej”.

procedure DoSomeAssembly; assembler;
asm
  // do some assembly
end;

Należy  jednak  uważać,  gdyż  procedura/funkcja  jest  wywoływana  zgodnie
z konwencjami ABI (Application Binary Interface). Z tego powodu blok asm jest znacznie
łatwiejszy w użyciu niż procedura asm. Na przykład w danym AVR zgodnie z ABI część
parametrów może być przekazywana przez rejestr i programista musi wziąć to pod uwagę.
Kompilator  nie  wstawia  żadnego  kodu  zarządzającego  aby  uzyskać  dostęp  do  tych
parametrów  jako  zmiennych  lokalnych,  jak  ma  to  miejsce  w  „normalnych”  procedurach
z dołączonym blokiem asm. Ponadto w przeciwieństwie do bloku asm programista musi sam
zadbać o używane rejestry. Tylko dla zmiennych lokalnych (jeśli są zdefiniowane) kompilator
wstawi kod zarządzający ramką stosu.

Interfejs  binarny  aplikacji  (ABI)  to  zbiór  zasad  określających  między  innymi
sposób wywoływania  podprogramów (procedur,  funkcji),  przekazywania  im argumentów
oraz odbierania zwracanej wartości. Zgodnie ze specyfikacją, z której korzysta FPC (https://
gcc.gnu.org/wiki/avr-gcc)  poszczególne  argumenty  (parametry)  funkcji  są  przekazywane
w kolejności od lewej do prawej z użyciem  rejestrów albo pamięci (jeśli zabraknie wolnych
rejestrów). Algorytm określający numer rejestru Rn odpowiadającego danemu parametrowi
jest następujący:

1. Przypisz liczbę 26 do n.
2. Jeśli rozmiar parametru jest nieparzysta liczbą bajtów zaokrągli go do najbliższej liczby
parzystej
3. Wynik odejmij od n.
4. Jeśli  n wynosi co najmniej 8 i rozmiar parametru jest większy od zera wówczas mniej
znaczący  (dolny)  bajt  parametru  jest  przekazywany  w  Rn a  następne  w  kolejnych  wg
rosnącej numeracji.
5. Jeśli nowy numer n jest mniejszy niż 8 lub rozmiar parametru jest równy zero, parametr
zostanie przekazany w pamięci.
6. Jeśli  bieżący parametr  jest  przekazywany w pamięci,  zatrzymaj procedurę -  wszystkie
kolejne argumenty również zostaną przekazane w pamięci.
7. W przeciwnym przypadku, jeśli pozostały argumenty bez kreślonych rejestrów, przejdź do
punktu 2.

W odniesieniu  do  wartości  zwracanych  przez  funkcje  obowiązuje  zasada,  iż  te
o  rozmiarze  od  1  do  8  bajtów  włącznie  będą  zwracane  w  rejestrach  a  te  o  rozmiarze
większym będą zwracane w pamięci. Rozmiary typów rekordowych (struktur) do 8 bajtów
są uzupełniane do następnej potęgi liczby 2 jeśli są zwracane w rejestrach.  Jeśli  wartość
zwracana przez funkcję nie może zostać zwrócona w rejestrach, kod ją wywołujący musi
przydzielić przestrzeń stosu i przekazać adres jako niejawny pierwszy argument (wskaźnik)
do funkcji wywoływanej. Parametry typu var (varargs) są przekazywane na stosie.

Przykład:

function func (a: char; b: UInt32): Int16;

1. Parametr  a zostanie przekazany w R24 (typ  char ma rozmiar  1 bajta więc  n zostanie
zaokrąglony do 2 i odjęte od 26).

2. Parametr  b zostanie  przekazany do R20, R21, R22 i  R23, z mniej  znaczącym bajtem
w R20 i bardziej znaczącym w R23 (UInt32 ma rozmiar 4 bajtów).
3. Wynik zostanie zwrócony w R24 i R25 (Int16 ma rozmiar 2 bajtów). 

9. Sekcje

Danych

Dane (zmienne) można umieścić w różnych sekcjach linkera za pomocą dyrektywy
section. Należy zachować ostrożność podczas umieszczania danych w sekcjach .progmem lub
.eeprom,  ponieważ  sekcje  te  nie  są  mapowane  na  przestrzeń  danych  dla  klasycznych
mikrokontrolerów AVR, a kompilator obecnie nie generuje poprawnego kodu do uzyskiwania
dostępu do tych sekcji. Przykładem może być umieszczenie zmiennej w sekcji  .noinit. Taka
zmienna  trafia  do  niezainicjowanej  pamięci  RAM,  więc  po  „miękkim”  (programowym)
restarcie  mikrokontrolera jej  wartość zostałaby zachowana. Uwaga: sekcja  .noinit wymaga
FPC w wersji 3.3.1, aby działać poprawnie.

Kodu

Obsługa  kojarzenia  informacji  sekcji  z  procedurami  i  funkcjami  została
zaimplementowana w rozwojowej wersji 3.3.1 FPC. Kod startowy został zmodyfikowany (jak
dotąd  tylko  dla  kontrolera/symulatora  avrsim)  tak,  aby  zachować  zgodność  z  konwencją
nazewnictwa sekcji avr-libc .initN:

-  punkt  wejścia  (ang.  entry  point) -  pierwsza  instrukcja  przechodzi  do  etykiety
__trampolines_end utworzonej przez linker. Po instrukcji skoku następuje tabela wektorów
przerwań. Jej kod jest inny dla każdego mikrokontrolera.
-  dane  programu  (ang.  progmem  data)  -  w  następnej  kolejności  umieszczane  są  dane
programu (o ile  zostały zdefiniowane).  Etykieta  __trampolines_end wskazuje na pierwszy
adres po tej lokalizacji.
.init0 - nieużywane, dostępne dla użytkownika 
.init1 - nieużywane, dostępne dla użytkownika 
.init2 - kod inicjujący rejestr zerowy i wskaźnik stosu 
.init3 - nieużywane, dostępne dla kodu użytkownika
.init4  -  kod  kopiujący  zainicjowane  zmienne  do  pamięci  RAM  i  zerujący  pamięć
niezainicjowanych zmiennych. 
.init4 do .init8 - nieużywane, dostępne dla użytkownika 
.init9 – przeskok do głównego kodu programu (ang. main)

Trzeba mieć na uwadze fakt, że kod finalizacji wygenerowany przez FPC następuje
bezpośrednio po reszcie kodu, przed początkiem sekcji  .finiN. Obecnie nie można zastąpić
kodu finalizacji wygenerowanego przez FPC ani wstawić własnego kodu za pośrednictwem
sekcji .fini, który zostałby wykonany po kodzie finalizacji.

program test;

// Store MCU startup status early and set to 0 so that e.g. watchdog resets
can be determined

var
  startupMCUSR: byte; section '.noinit';

// This should be marked noreturn since it should fall through into the 
next init section
procedure saveMCUSR; noreturn; section '.init1';
begin
  startupMCUSR := MCUSR;
  MCUSR := 0;
end;

begin
  if startupMCUSR = 1 then
    writeln('Reset cause: Power-on');
end.

Wyniki  kompilacji  kodu modułów i  programu są  umieszczane  w tzw.  sekcjach
pamięci,  zawierających  specyficzne  dla  siebie  dane.  Dzięki  istnieniu  sekcji  linker  może
„poukładać”  razem  informacje  z  różnych  plików  wynikowych,  zgodnie  z  ich
przeznaczeniem. Linker  avr-ld obsługuje kilka rodzajów sekcji należących do określonych
regionów pamięci mikrokontrolera, tj. kodu (text), danych (data), pamięci EEPROM, bajtów
bezpieczników  (fuse),  lock-bitów,  sygnatury  oraz  sygnatury  użytkownika,  przy  czym
niektóre regiony składają się z więcej niż jednej sekcji a poszczególne sekcje mogą zawierać
podsekcje. Nazwy sekcji rozpoczynają się od kropki. Zwykle w skryptach konfiguracyjnych
linkera (tworzone przez FPC „w tle” na czas kompilacji) definiowane są następujące sekcje:

1. .text  – zawiera kod wykonywalny oraz dane typu progmem (podsekcja .progmem);  ta
sekcja jest następnie podzielona na podsekcje .initN (N od 0 do 9) oraz .finiN (N od 9 do 0) 
2. .data – zawiera statyczne dane
3.  .bss –  niezainicjowane,  zerowane  zmienne  globalne  lub  statyczne;  ta  sekcja  może
zawierać dodatkową podsekcję .noinit – niezainicjowanych danych statycznych, które nie są
zerowane podczas startu programu
4. .eeprom – dane w pamięci EEPROM
5. .fuse – bajty bezpieczników
6. .lock – bajty lock-bitów
7. .signature – bajty sygnatury

Ponadto kompilator (w zależności od ustawień) może tworzyć dodatkowe sekcje np.
przeznaczone  dla  debugera  (.debug_).  Szczegółowe  informacje  na  temat  sekcji  pamięci
biblioteki avr-libc można uzyskać tutaj:

https://www.datsi.fi.upm.es/docencia/DMC/avr-libc-user-manual-2.2.0/mem_sections.html
https://www.nongnu.org/avr-libc/user-manual/mem_sections.html

* * *

Jeśli Czytelnik dotarł do tego miejsca, może to uznać za sukces! Dobra wiadomość
jest taka, że pełna znajomość tych wszystkich informacji nie jest potrzebna do korzystania
z niniejszego kursu. Kolejne odcinki zaczną się od zagadnień prostych, nie wymagających
wyrafinowanego kodu a każde nowe konstrukcje stosowane w programach będą wyjaśniane
na bieżąco. Niniejszą część należy raczej traktować jako miejsce do którego w razie potrzeby
można wrócić. 

Andrzej Karwowski

Nobody is sperfect – znalazłeś/-aś błąd lub nieścisłość, masz pytanie bądź sugestię – napisz na
adres: ackarwow@gmail.com.

Podziękowania

Dziękuję użytkownikowi Chivo z forum www.elektroda.pl za wskazanie braku opisu
modułów i bibliotek w rozdziale 1. Podziękowania należą się również użytkownikowi sp3ots
tegoż  forum  za  pytanie  „dlaczego  Pascal?”,  na  które  starałem  się  odpowiedzieć  oraz
wskazanie zagadnień dotyczących operacji na bitach rejestrów wartych rozwinięcia w tekście.

